

XPathAPI class

ActionScript Class Name mx.xpath.XPathAPI

The XPathAPI class allows you to do simple XPath searches within Macromedia Flash. This can be very useful for searching XML packets based on node names and attribute values. In other words, you can quickly find nodes and attributes in an XML document using the XPathAPI methods.

To do XPath searches within Flash, you must first include the XPathAPI class in your Flash library by adding the DataBindingClasses component (if it hasn't been added already). If you've already set up bindings, this class may have been included automatically; otherwise, you must select the class from the common libraries (Window > Common Libraries > Classes). From the Classes.fla library panel, you can drag a copy of the DataBindingClasses component into your current Flash document's library. You then import the XPathAPI class by typing `import mx.xpath.XPathAPI`, or by using the class's fully qualified class name as a prefix when accessing the class methods (`mx.xpath.XPathAPI.method_name`).

Flash supports the following path expressions:

Type	Example
Absolute path	<code>/item/title</code>
Relative path	<code>title</code> if the context node is <code><item></code>
wildcard (*)	<code>/*/title</code> retrieves all <code><title></code> elements, whatever the parent node is
predicate expressions using <code>=</code> , <code>AND</code> , or <code>OR</code>	<code>/item/title[@version='current']</code> or: <code>/item/title[@version='current' AND @post='today']</code>

Method summary for the XPathAPI class

The following table lists the available methods of the XPathAPI class.

Method	Description
<code>XPathAPI.getEvalString()</code>	For a specified node, gets the corresponding evaluation string.
<code>XPathAPI.selectNodeList()</code>	Retrieves all the node values of a specified node level.
<code>XPathAPI.selectSingleNode()</code>	Retrieves the node value of the first node at a specified node level.
<code>XPathAPI.setNodeValue()</code>	Sets the value of a specified node or attribute.

XPathAPI.getEvalString()

Availability

Flash Player 6 (6.0.79.0).

Edition

Flash MX 2004.

Usage

`XPathAPI.getEvalString(node, path)`

Parameters

node An XML node; a reference to the DOM node that is the parent node from which the statement executes.

path A string identifying the path through the XML node hierarchy.

Returns

A string that represents the code required to access the value specified by the path parameter from the parent node.

Description

Method; for a specified node, gets the corresponding evaluation string.

Example

The following example displays the corresponding evaluation string used for a specified node in the Output panel:

```
import mx.xpath.XPathAPI;
var rssfeed_xml:XML = new XML();
rssfeed_xml.ignoreWhite = true;
rssfeed_xml.onLoad = function(success:Boolean) {
    trace("onload...");
    if (success) {
        trace("success...");
        // Retrieve all title notes within /rss/channel/item.
        var thePath_str:String = "/rss/channel/item";
        // Get the evaluation string used for the search.
        var myEvalString_str:String = XPathAPI.getEvalString(this.firstChild,
thePath_str);
        trace("getEvalString: " + myEvalString_str);
    } else {
        trace("error loading XML");
    }
};
rssfeed_xml.load("http://www.flash-mx.com/news/index.xml");
```

XPathAPI.selectNodeList()

Availability

Flash Player 6 (6.0.79.0).

Edition

Flash MX 2004.

Usage

`XPathAPI.selectNodeList(node, path)`

Parameters

node An XML node; a reference to the DOM node that is the parent node from which the statement executes.

path A string identifying the path through the XML node hierarchy.

Returns

An array of nodes matching the given XPathAPI statement.

Description

Method; retrieves all the node values of a specified node level.

Example

The following example stores all the title node values into an array, and then displays the list in the Output panel:

```
import mx.xpath.XPathAPI;
var rssfeed_xml:XML = new XML();
rssfeed_xml.ignoreWhite = true;
rssfeed_xml.onLoad = function(success:Boolean) {
    trace("onload...");
    if (success) {
        trace("success...");
        // Retrieve all title notes within /rdf:RDF/item/title.
        var thePath_str:String = "/rdf:RDF/item/title";
        var title_array:Array = XPathAPI.selectNodeList(this.firstChild,
thePath_str);
        for (var i:Number = 0; i < title_array.length; i++) {
            trace(title_array[i].firstChild.nodeValue);
        }
    } else {
        trace("error loading XML");
    }
};
rssfeed_xml.load("http://weblogs.macromedia.com/dev_center/index.rdf");
```

XPathAPI.selectSingleNode()

Availability

Flash Player 6 (6.0.79.0).

Edition

Flash MX 2004.

Usage

`XPathAPI.selectSingleNode(node, path)`

Parameters

node An XML node; a reference to the DOM node that is the parent node from which the statement executes.

path A string identifying the path through the XML node hierarchy.

Returns

The XML node found for the specified path and node, or null if not found.

Description

Method; returns a single node from the given node and XPath parameters. If the path specified will result in multiple nodes returned, this method will return only the first one found.

Example

The following example finds the first title node value within the item node, and then displays the value in the Output panel:

```
import mx.xpath.XPathAPI;
var rssfeed_xml:XML = new XML();
rssfeed_xml.ignoreWhite = true;
rssfeed_xml.onLoad = function(success:Boolean) {
    trace("onload...");
    if (success) {
        trace("success...");
        // Retrieve first title node within /rdf:RDF/item/title.
        var thePath_str:String = "/rdf:RDF/item/title";
        var titleNode_str:String = XPathAPI.selectSingleNode(this.firstChild,
thePath_str);
        trace(titleNode_str.toString());
    } else {
        trace("error loading XML");
    }
};
rssfeed_xml.load("http://weblogs.macromedia.com/dev_center/index.rdf");
```

XPathAPI.setNodeValue()

Availability

Flash Player 6 (6.0.79.0).

Edition

Flash MX 2004.

Usage

`XPathAPI.setNodeValue(node, path, newValue)`

Parameters

node An XML node; a reference to the DOM node that is the parent node from which the statement executes.

path A string identifying the path through the XML node hierarchy.

newValue A string containing the new value for the node or attribute.

Returns

The number of nodes updated.

Description

Method; sets the value of the node or attribute identified in the *path* string to the value of the *newValue* parameter.

Example

The following example changes the node values for the specified node *path*, and then displays a summary of the changes in the Output panel:

```
import mx.xpath.XPathAPI;
var rssfeed_xml:XML = new XML();
rssfeed_xml.ignoreWhite = true;
rssfeed_xml.onLoad = function(success:Boolean) {
    trace("onload...");
    //trace(rssfeed_xml);
    if (success) {
        var i:Number;
        // Retrieve all title notes within /rss/channel/item.
        var thePath_str:String = "/rss/channel/item/dc:creator";
        var creator_array:Array = XPathAPI.selectNodeList(this.firstChild,
thePath_str);
        trace("before:");
        for (i = 0; i < creator_array.length; i++) {
            trace("\t" + i + ". " + creator_array[i].firstChild.nodeValue);
        }
        var updatedNodes_num:Number = XPathAPI.setNodeValue(this.firstChild,
thePath_str, "dave");

        trace(" ***** " + updatedNodes_num + " nodes updated ***** ");

        var creator_array:Array = XPathAPI.selectNodeList(this.firstChild,
thePath_str);
        trace("after:");
        for (i = 0; i < creator_array.length; i++) {
            trace("\t" + i + ". " + creator_array[i].firstChild.nodeValue);
        }
    } else {
        trace("error loading XML");
    }
};
rssfeed_xml.load("http://www.flash-mx.com/news/index.xml");
```